

CS3388B: Lecture 16

March 28, 2023

16 Geometry and Tessellation Shaders: Part II

16.1 Tessellation Control Shaders

A TCS is invoked on a single patch of vertices.

- The number of vertices N in a patch is specified by:

```
1 layout (vertices= N ) out;
```

- **The TCS is invoked once per vertex per patch.** Which vertex a particular invocation corresponds to is held in the `gl_InvocationID` built-in input variables.
- All inputs from vertex shaders to the TCS are aggregated into arrays, based on the size of the input patch N . User-defined outputs from the vertex shader can be accessed like:

```
1 in vec3 color_vs[];  
2 in float someFloat_vs[];
```

- The built-in outputs from the vertex shader can be accessed like:

```
1 gl_in[gl_InvocationID].gl_Position;
```

- The TCS should write user-defined per-vertex data out to an array using `gl_InvocationID` as index:

```
1 out vec3 color_tcs[];  
2 //...  
3 color_tcs[gl_InvocationID] = color_vs[gl_InvocationID];
```

- The TCS should write per-vertex data out to an array using `gl_InvocationID` as index:

```
1 out vec3 color_tcs[];  
2 //...  
3 color_tcs[gl_InvocationID] = color_vs[gl_InvocationID];
```

- The TCS must write to `gl_TessLevelOuter[]` and `gl_TessLevelInner[]` to control the tessellation levels. The number of indices to use for each outer and inner depends on the

abstract patch type defined in the Tessellation evaluation shader.

- For triangles: 3 outer levels, 1 inner level;
- For quads: 4 outer levels, 2 inner levels;
- For isolines: 2 outer levels, 0 inner levels;

- The TCS must write to `gl_out[] .gl_Position`.

```
1 //A Tessellation Control Shader
2 #version 400
3
4 layout (vertices = 3) out;
5
6 in vec3 color_vs[];
7 out vec3 color_tcs[];
8
9 uniform float outerTess;
10 uniform float innerTess;
11
12 void main() {
13     gl_out[ gl_InvocationID ].gl_Position = gl_in[ gl_InvocationID ].
        gl_Position;
14
15     color_tcs[gl_InvocationID] = color_vs[gl_InvocationID];
16
17     gl_TessLevelOuter[0] = outerTess;
18     gl_TessLevelOuter[1] = outerTess;
19     gl_TessLevelOuter[2] = outerTess;
20     gl_TessLevelInner[0] = innerTess;
21 }
```

Note: `gl_InvocationID` is the index within the patch of the incoming vertex. If your vertex buffer has:

```
1 -1, 1, 0,
2 1, 1, 0,
3 0, 0, 0
```

Then vertex (-1, 1, 0) has `gl_InvocationID 0` and `gl_in[0]` corresponds to whatever the vertex shader outputs when processing the vertex (-1,1,0).

16.2 Tessellation Evaluation Shader

Depending on the tessellation levels specified by the tessellation control shader, the fixed-function tessellation primitive generation produces extra vertices from the input patch vertices and the abstract primitives they define. The type of abstract primitive is defined in the Tessellation Evaluation Shader (TES).

Moreover, the vertices created during tessellation must have vertex attributes assigned to them. This is the job of the TES. It acts as a secondary vertex shader for the tessellation-created vertices.

Vertex attributes output from the vertex shader are not automatically interpolated during tessellation primitive generation.

The TES must itself compute the correct vertex attributes, including position, for each of the newly generated vertices. The TES is invoked once per new vertex. It must compute the:

1. Vertex position (in clip space or whatever is expected by the geometry shader).
2. Any user-defined vertex attributes expected by the geometry shader or fragment shader.

In particular, the TES has the following specifications:

1. The abstract patch type and *how* tessellation should be applied is defined by the layout parameter:

```
1 layout (triangles, equal_spacing) in;
```

The abstract patch types are triangles, quads, and isolines. The spacing types are `equal_spacing`, `fractional_even_spacing`, `fractional_odd_spacing`.



2. All inputs from TCS to the TES are aggregated into arrays, based on the number of vertices in the abstract patch type.

User-defined inputs and built-in inputs are accessed just as the TES, but using the vertex indices of the abstract patch rather than `gl_InvocationID`.

3. All outputs of the TES are per-vertex and are thus **not** arrays. This includes the normal vertex shader outputs like `gl_Position`, and any other vertex attribute expected by the geometry/fragment shader.
4. When the abstract patch is a quad, the outputs of the TES are computed **similar to texture coordinates**, using a normalized 2D point to interpolate between the 4 corners of the quad. It uses the first two indices from `vec3 gl_TessCoord`.
5. When the abstract patch is a triangle, the outputs of the TES are computed using **barycentric coordinates** and three indices from `vec3 gl_TessCoord`.

16.2.1 Barycentric Coordinates

Without going into the math of it, **barycentric coordinates** give the position within a triangle as a 3D vector. The vector gives the **weights in a weighted sum** of the three corners.



```

1  //A Tessellation Evaluation Shader
2  #version 400
3
4  layout (triangles, equal_spacing) in;
5
6  uniform mat4 MVP;
7
8  in vec3 color_tcs[];
9  out vec3 color_gs;
10
11 void main() {
12
13     vec4 p1= gl_in[0].gl_Position;
14     vec4 p2= gl_in[1].gl_Position;
15     vec4 p3= gl_in[2].gl_Position;
16
17     gl_Position = vec4(0.0);
18     gl_Position += gl_TessCoord.x *p1;
19     gl_Position += gl_TessCoord.y *p2;
20     gl_Position += gl_TessCoord.z *p3;
21
22     color_gs = vec3(0.0);
23     color_gs += gl_TessCoord.x *color_tcs[0];
24     color_gs += gl_TessCoord.y *color_tcs[1];
25     color_gs += gl_TessCoord.z *color_tcs[2];
26 }

```

When using a Tessellation evaluation shader the outputs must match the inputs of the geometry/fragment shader which is next in the render pipeline.

16.3 Geometry Shaders with Tessellation Shaders: A Use Case

One possibly annoying part of tessellation shaders is their implicit interpolation of vertex attributes. Since each vertex attribute output from a TES is computed using `gl_TessCoords`, the vertices output from tessellation may “lose” the fact they are now independent primitives generated from pre-existing ones.

Consider defining a vertex attribute for each corner of a tessellation-generated triangle as:

```
1 vec3 attrib1(1, 0, 0);
2 vec3 attrib2(0, 1, 0);
3 vec3 attrib3(0, 0, 1);
```

Since each new vertex is computing using `gl_TessCoord`, it is impossible (or at least very hard) to determine which vertex corresponds to which discrete primitive.

But, this is exactly the job of the geometry shader! To process per-primitive attributes. Once a tessellation evaluation shader has created vertices and vertex attributes, those go into the geometry shader as if they are full-fledged primitives. The geometry shader does not know if they are user-defined vertices or tessellation-generated vertices.

So long as the input primitive type of the geometry shader matches the abstract patch type of a Tessellation Evaluation Shader, you are in promising territory.

```
1 //A Geometry Shader
2 #version 400
3
4 layout (triangles) in;
5 layout (triangle_strip, max_vertices=3) out;
6
7 in vec3 color_tes[];
8 flat out vec3 color_gs;
9 out vec3 triDistance;
10
11 void main() {
12     gl_Position = gl_in[0].gl_Position;
13     color_gs = color_tes[0];
14     triDistance = vec3(1, 0, 0);
15     EmitVertex();
16
17     gl_Position = gl_in[1].gl_Position;
18     color_gs = color_tes[1];
19     triDistance = vec3(0, 1, 0);
20     EmitVertex();
21
22     gl_Position = gl_in[2].gl_Position;
23     color_gs = color_tes[2];
24     triDistance = vec3(0, 0, 1);
25     EmitVertex();
26     EndPrimitive();
27 }
```

In the above geometry shader, the `triDistance` attribute will be interpolated across each fragment generated by this primitive during rasterization. Thus, this `triDistance` corresponds to how close the generated fragment is to the edges of the primitive. This can be taken advantage of in the fragment shader to generate edges to the tessellation.

```
1 #version 400
2
3 flat in vec3 color_gs;
4 in vec3 triDistance;
5
6 out vec4 color;
7
8 void main(){
9
10     float d1 = min(min(triDistance.x, triDistance.y), triDistance.z);
11     if (d1 < 0.02) {
12         color = vec4(0, 0, 0, 1);
13     } else {
14         color = vec4(color_gs, 1.0);
15     }
16 }
```

16.4 The Final Step

Finally, we need a couple of things in our main C/C++ program to make our shaders work correctly.

- `glPatchParameteri(GL_PATCH_VERTICES, N);` must be called, where `N` is the number of vertices in the TCS.
- you must use `GL_PATCHES` as the primitive type in your calls to `glDrawArrays` or `glDrawElements`